

Self-Teaching Package

Version 2012

SAS version - LWS



1. Differences in Concepts of Net Worth (SAS software)

Goal

Estimates of net worth can differ substantially depending on the wealth measure you use. In this exercise, you will begin to familiarize yourself with the summary measures in LWS and determine the differences in portfolio compositions for the whole population using two different definitions of net worth. In the final part of the exercise, you will compare **lis_dpi** and **nw2** by the quintile groups.

Activity

Calculate summary statistics (mean and median) for two LWS net worth concepts, **nw1** and **nw2**, as defined in the *LWS Variable Definition List* and *Construction of Aggregate Variables*. Determine the differences in portfolio compositions for these two measures. Second, divide the sample into the quintile groups by **nw2** and calculate the mean and median for **nw2** and **lis_dpi**. Perform the analysis on the United States SCF 2000, 2003, and 2006.

Guidelines

- Don't forget to change the project in your job submission panel to LWS.
- Use the *LWS Variable Definition List* and *Construction of Aggregate Variables* to identify the components of **nw1** and **nw2**.
- For the first part of the exercise calculate the means and medians for the two net worth measures.
- For the second part of the exercise calculate the means of the components and take the appropriate ratios to find the shares.
- For business holdings, use the measure for business equity (**be**).
- Please note that the documentation for US 2003 and US 2006 is the same as for US 2000.
- Since there are 2 different surveys for US 2000 (PSID and SCF), therefore to access US SCF surveys you need to call the data as follows: **uss00**, **uss03**, and **uss06**.
- Remember to use the weights (**wgt**)
- To calculate the quintile groups in SAS you need to use the **PROC UNIVARIATE** procedure, the **CALL SYMPUT** routine, and a **_NULL_** data step, as already demonstrated in the exercises 10, 11, and 13.

Use your results to fill in the following tables:

Table 1

	United States 2000	United States 2003	United States 2006
Net worth (definition 1)			
Mean			
Median			
Net worth (definition 2)			
Mean			
Median			

Table 2

	US 2000		US 2003		US 2006	
	<i>nw1</i>	<i>nw2</i>	<i>nw1</i>	<i>nw2</i>	<i>nw1</i>	<i>nw2</i>
Total Financial Assets (definition 1)						
Total Non-financial Assets (definition 1)						
Business Equity						
Total Assets (sum of the 2 or 3 lines above)						
Debt						
Net worth (total assets – debt)						

Table 3

			US 2000	US 2003	US 2006
1st quintile	<i>nw2</i>	Mean			
		Median			
	<i>lis_dpi</i>	Mean			
		Median			
2nd quintile	<i>nw2</i>	Mean			
		Median			
	<i>lis_dpi</i>	Mean			
		Median			
3rd quintile	<i>nw2</i>	Mean			
		Median			
	<i>lis_dpi</i>	Mean			
		Median			
4th quintile	<i>nw2</i>	Mean			
		Median			
	<i>lis_dpi</i>	Mean			
		Median			
5th quintile	<i>nw2</i>	Mean			
		Median			
	<i>lis_dpi</i>	Mean			
		Median			

Program

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER  
LABEL LS=MAX PS=MAX;
```

```
%MACRO net;  
DATA net;  
    SET &&&pi.w (KEEP=wgt nw1 nw2 tfa1 tnf1 td be ba lis_dpi);  
RUN;  
PROC MEANS DATA=net MEAN MEDIAN;  
    VAR nw1 nw2 tfa1 tnf1 td be ba;  
    WEIGHT wgt;  
RUN;  
PROC UNIVARIATE NOPRINT DATA=net;  
    VAR nw2;  
    WEIGHT wgt;  
    OUTPUT OUT=quintile pctlpts=20 40 60 80 pctlpre=pct;  
RUN;  
DATA _NULL_;  
    SET quintile;  
    CALL SYMPUT('q1_nw2',pct20) ;  
    CALL SYMPUT('q2_nw2',pct40) ;  
    CALL SYMPUT('q3_nw2',pct60) ;  
    CALL SYMPUT('q4_nw2',pct80) ;  
RUN;  
DATA net;  
    SET net;  
    IF nw2 <= &q1_nw2 then nw2_quint=1;  
    ELSE IF nw2 <= &q2_nw2 THEN nw2_quint=2;  
    ELSE IF nw2 <= &q3_nw2 THEN nw2_quint=3;  
    ELSE IF nw2 <= &q4_nw2 THEN nw2_quint=4;  
    ELSE nw2_quint=5;  
RUN;  
PROC UNIVARIATE NOPRINT DATA=net;  
    VAR lis_dpi ;  
    WEIGHT wgt;  
    OUTPUT OUT=quintile pctlpts=20 40 60 80 pctlpre=pct;  
RUN;  
DATA _NULL_;  
    SET quintile;  
    CALL SYMPUT('q1_dpi',pct20) ;  
    CALL SYMPUT('q2_dpi',pct40) ;  
    CALL SYMPUT('q3_dpi',pct60) ;  
    CALL SYMPUT('q4_dpi',pct80) ;  
RUN;  
DATA net;  
    SET net;  
    IF lis_dpi <= &q1_dpi then dpi_quint=1;  
    ELSE IF lis_dpi <= &q2_dpi THEN dpi_quint=2;
```

```
ELSE IF lis_dpi <= &q3_dpi THEN dpi_quint=3;
ELSE IF lis_dpi <= &q4_dpi THEN dpi_quint=4;
ELSE dpi_quint=5;
RUN;
```

```
PROC MEANS DATA=net MEAN MEDIAN;
CLASS nw2_quint;
VAR nw2;
WEIGHT wgt;
```

```
RUN;
```

```
PROC MEANS DATA=net MEAN MEDIAN;
CLASS dpi_quint;
VAR lis_dpi;
WEIGHT wgt;
```

```
RUN;
```

```
%MEND net;
```

```
%LET pi = uss00;
```

```
%net
```

```
%LET pi = uss03;
```

```
%net
```

```
%LET pi = uss06;
```

```
%net
```

Results

Table 1.

	United States 2000	United States 2003	United States 2006
Net worth (definition 1)			
Mean	<i>212,204</i>	<i>250,477</i>	<i>295,917</i>
Median	<i>43,110</i>	<i>47,475</i>	<i>60,300</i>
Net worth (definition 2)			
Mean	<i>287,056</i>	<i>337,950</i>	<i>423,445</i>
Median	<i>48,120</i>	<i>53,650</i>	<i>67,200</i>

Table 2.

	US 2000		US 2003		US 2006	
	<i>nw1</i>	<i>nw2</i>	<i>nw1</i>	<i>nw2</i>	<i>nw1</i>	<i>nw2</i>
Total Financial Assets (definition 1)	<i>37.9%</i> <i>(102,299)</i>	<i>29.6%</i> <i>(102,299)</i>	<i>30.2%</i> <i>(101,436)</i>	<i>24%</i> <i>(101,436)</i>	<i>28.8%</i> <i>(114,980)</i>	<i>21.8%</i> <i>(114,980)</i>
Total Non-financial Assets (definition 1)	<i>62.1%</i> <i>(167,932)</i>	<i>48.7%</i> <i>(167,932)</i>	<i>69.8%</i> <i>(234,090)</i>	<i>55.3%</i> <i>(234,090)</i>	<i>71.2%</i> <i>(283,754)</i>	<i>53.9%</i> <i>(283,754)</i>
Business Equity	-----	<i>21.7%</i> <i>(74,852)</i>	-----	<i>20.7%</i> <i>(87,473)</i>	-----	<i>24.3%</i> <i>(127,528)</i>
Total Assets (sum of the 2 or 3 lines above)	<i>100%</i> <i>(270,231)</i>	<i>100%</i> <i>(345,083)</i>	<i>100%</i> <i>(335,526)</i>	<i>100%</i> <i>(422,999)</i>	<i>100%</i> <i>(398,734)</i>	<i>100%</i> <i>(526,262)</i>
Debt	<i>21.5%</i> <i>(58,028)</i>	<i>16.8%</i> <i>(58,028)</i>	<i>25.3%</i> <i>(85,049)</i>	<i>20.1%</i> <i>(85,049)</i>	<i>25.8%</i> <i>(102,817)</i>	<i>19.5%</i> <i>(102,817)</i>
Net worth (total assets – debt)	<i>78.5%</i> <i>(212,203)</i>	<i>83.2%</i> <i>(287,056)</i>	<i>74.7%</i> <i>(250,477)</i>	<i>79.9.2%</i> <i>(337,950)</i>	<i>74.2%</i> <i>(295,917)</i>	<i>80.5%</i> <i>(423,445)</i>

Table 3.

			US 2000	US 2003	US 2006
1st quintile	<i>nw2</i>	Mean	-10,111	-12,660	-16,441
		Median	-4,580	-6,300	-7,770
	<i>lis_dpi</i>	Mean	5,876	6,887	3,590
		Median	8,965	9,641	10,742
2nd quintile	<i>nw2</i>	Mean	6,187	7,639	8,867
		Median	4,500	4,800	5,700
	<i>lis_dpi</i>	Mean	19,074	21,272	22,539
		Median	19,033	21,145	22,614
3rd quintile	<i>nw2</i>	Mean	49,211	56,219	70,473
		Median	48,120	53,650	67,200
	<i>lis_dpi</i>	Mean	29,409	32,664	34,686
		Median	29,006	32,409	34,390
4th quintile	<i>nw2</i>	Mean	145,785	172,156	204,735
		Median	136,500	162,600	200,000
	<i>lis_dpi</i>	Mean	44,861	49,580	53,384
		Median	44,715	49,033	52,470
5th quintile	<i>nw2</i>	Mean	1,246,719	1,468,949	1,853,491
		Median	504,000	601,300	663,980
	<i>lis_dpi</i>	Mean	116,591	168,417	171,700
		Median	77,982	88,716	93,237

2. Asset Participation (SAS software)

Goal

The goal of this exercise is to become familiar with different types of assets in the LWS data and to compare asset participation of the elderly with the population as a whole.

Activity

Calculate participation in the three assets (deposit accounts, stocks, investment real estate, business assets/equity) for the total population, and the elderly population in Finland in 1998, Italy 2002 and Sweden 2002.

Use the information from your output and/or the documentation to answer the following questions:

1. In which country are households more inclined to have risky portfolios?

2. What happens to deposit accounts as the population ages?

Guidelines

- Use the *LWS Quick Reference Guide* to help you with the job submission.
- Identify the wealth variables needed to calculate the participation rates using the documentation to check whether each of these components exists in each of these countries.
- Create dummy variables for each of the wealth components to indicate that a household is holding a particular asset. Here is an example of the dummy generation:

```
dda = 0 ;  
IF ((da > 0 ) OR (da = .)) THEN dda = 1 ;
```

- For business holdings, use the measure for business equity, if available. Otherwise, use business assets. In order to do this, you will need to check the country-specific documentation for the availability of business assets, business debt, and business equity.
- When measuring assets of the elderly population, define elderly households as those with a head or spouse 65 years of age or older.

Use your results to fill in the following tables:

Total population	Finland 1998	Italy 2002	Sweden 2002
Deposit Accounts			
Stocks			
Investment Real Estate			
Business Assets			
Business Equity			

Elderly Population 65+	Finland 1998	Italy 2002	Sweden 2002
Deposit Accounts			
Stocks			
Investment Real Estate			
Business Assets			
Business Equity			

Program

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER  
LABEL LS=max PS=max ;
```

```
/**-----**/
```

```
/** MACRO DECLARATION **/
```

```
/**-----**/
```

```
%MACRO avg ;  
  PROC MEANS DATA=tmp MEAN;  
  &where ;  
  VAR dda dst dir dbe dba ;  
  WEIGHT wgt ;  
  RUN;  
%MEND avg ;
```

```
%MACRO asset ;  
  DATA tmp ;  
  SET &&&pi.w (KEEP=ctry wgt ageh ages da st ir ba be td nw1 nw2);  
  dda = 0 ;  
  IF ((da > 0 ) OR (da = .)) THEN dda = 1 ;  
  dst = 0 ;  
  IF ((st > 0 ) OR (st = .)) THEN dst = 1 ;  
  dir = 0 ;  
  IF ((ir > 0 ) OR (ir = .)) THEN dir = 1 ;  
  dbe = 0 ;  
  IF (be > 0 ) THEN dbe = 1 ;  
  dba = 0 ;  
  IF (ba > 0 ) THEN dba = 1 ;  
  eld = 0 ;  
  IF ((ageh>=65) OR (AGES >=65)) THEN eld = 1 ;  
  RUN;  
  TITLE1 "Country: " &pi ;  
  TITLE2 "TOTAL POPULATION";  
  %LET where = ;  
  %avg ;  
  TITLE2 "ELDERLY";  
  %LET where = WHERE (eld = 1) ;  
  %avg ;  
%MEND asset;
```

```
/**-----**/
```

```
/** START PROGRAM **/
```

```
/**-----**/
```

```
%LET pi=fi98;  
%asset  
%LET pi=it02;  
%asset  
%LET pi=se02;  
%asset
```

Results

Total population	Finland 1998	Italy 2002	Sweden 2002
Deposit Accounts	<i>90.7</i>	<i>80.7</i>	<i>58.5</i>
Stocks	<i>32.9</i>	<i>10.1</i>	<i>36.3</i>
Investment Real Estate	<i>26.9</i>	<i>21.8</i>	<i>13.6</i>
Business Assets	---	<i>16.2</i>	<i>7.5</i>
Business Equity	---	<i>15.5</i>	---

Elderly Population 65+	Finland 1998	Italy 2002	Sweden 2002
Deposit Accounts	<i>88.4</i>	<i>72.9</i>	<i>70.2</i>
Stocks	<i>28.9</i>	<i>6.1</i>	<i>35.8</i>
Investment Real Estate	<i>29.9</i>	<i>19.5</i>	<i>14.6</i>
Business Assets	---	<i>4.8</i>	<i>8.3</i>
Business Equity	---	<i>4.7</i>	---

Comments

- Finland has a higher proportion of investments in stocks and real estate, but they also have a high investment in deposit accounts. In Sweden, investment is also high, but deposit accounts are lower, which suggests a portfolio with a riskier balance.
- Except in Sweden, deposit accounts are lower, suggesting a spending of funds as individuals age. In Sweden, however, deposits rise after 65, which may mean healthy retirement programs and/or a decrease in spending in later years